

Extending Meta-Casanova with Modules and Type Operators

Douwe van Gijn

supervised by
Dr. Giuseppe Maggiore

2016-02-21

Abstract

Meta-Casanova is a functional and declarative language designed for writing compilers. It is still a language in development and is only simply-typed. In this paper, we extend the language with type constructors and a module system to allow abstractions like generic monad transformers.

This paper is one of three papers written for the Meta-Casanova minor. The focus of this paper is about the type system and the modules we have developed during the minor. To learn more about the syntax and semantics of Meta-Casanova, refer to the paper by Jarno Holstein[1]. To learn about the evolution of the standard library of Meta-Casanova and its applications, refer to the paper by Louis van der Burg[2].

Contents

1 Introduction

2 The Problem

3 The Idea

4 Details

- 4.1 Meta-Casanova 3
- 4.2 TypeFunc Details 4
- 4.3 Module Details 5

5 Conclusions

1 Introduction

A strong type system is essential for a modern language. It enables powerful abstractions like monad transformers to happen at compile-time, making it ideal for a fast compiler language.

Because the focus of Meta-Casanova is on writing compilers, we begin with a brief overview of monadic parsing. We will discover that parser monads can be constructed from more elementary monads, and that there is a simple way to combine these monads into a parser monad. We will then find that we need a stronger typesystem than is currently supported by Meta-Casanova and that we need to extend it with Modules and Type Operators. The rest of the paper will go in detail on the design of this type system.

2 The Problem

- 1 The goal of this paper is to extend Meta-Casanova's typesystem to the point where it can handle monad-transformers. To see why, we begin with a brief overview of monadic parsing.

Monadic Parsing

6 Parser monads are a useful tool for writing compilers. They are not only useful in the lexing stage, as they can be used for interpreting intermediate representations.

The best aspect about monadic parsing is that it is *composable*. It is possible to combine multiple parsers into a single parser. For example: there might be a parser (`digit`) that can parse a single digit, and a parser combinator (`repeat`) that can apply a parser multiple times until it fails. It is then trivial to combine these into a new parser that parses multiple digits.

```
let digits = repeat digit
```

A parser is a function that takes an input stream and a context and returns a result. A generic implementation may look like this.

```
type Parser<'char, 'ctxt, 'result> =  
  List<'char> * 'ctxt  
  -> Result<'char, 'ctxt, 'result>
```

The Result contains —if the parser was successful— the value returned by

the parser, the modified context, and the new input stream position.

```
type Result<'char,'ctxt,'result> =
  | Done of 'result*List<'char>*'ctxt
  | Error of string
```

Monad Transformers

It may not be obvious, but the parser monad can be constructed from more primitive monads. The input stream and the context are represented as a state-monad, and the result is an either-monad.

```
type Parser<'char,'ctxt,'result> =
  State<List<'char>*'ctxt,
    Either<'result,'error>>
```

Combining monads by hand is tedious and non-trivial. Luckily there is a way to automate the process: monad transformers.[3, Chapter 18]

Monad transformers are *type operators* (type-level functions) that takes a monad, and returns a combined monad. For example: if we pass the identity monad to the state monad transformer, we get a state monad.

Towards F_ω

In order to have type operators, we need higher-order polymorphism. The formal name for a type system that supports higher-order polymorphism, System- F_ω . [4, Chapter 30]

3 The Idea

In order to provide the typesystem we require, additional syntax is needed. We need a way to define type operators and a way to create *kinds*.

Kinds

Kinds are to types as types are to terms.[4, Chapter 30] They are “one level up” from types. The most used

kind is $*$ (pronounced “type”) and represents all proper types. Type operators are indicated with \Rightarrow to differentiate them from type-signatures (\rightarrow). For example a few types with their respective kinds:

```
int          *
float        *
pair         *  $\Rightarrow$  *  $\Rightarrow$  *
pair int     *  $\Rightarrow$  *
pair float int *
pair pair    error
```

`pair pair` is not a valid type, because type `pair` has kind $* \Rightarrow * \Rightarrow *$. It only accepts two types with kind $*$. Kinds are necessary to prevent these nonsensical types from typechecking.

Kinds -unlike types- have no run-time representation. They don’t need it, they are only executed at compile-time to check the validity of the types.

This property of kinds can be used to compute arbitrary expressions at compile-time. In Meta-Casanova, these expressions are called `TypeFuncs`, as they operate on types.

Modules

Modules in Meta-Casanova are a compile-time mechanism that allows the definition of interfaces. It re-uses `TypeFuncs` to declare new kinds.

To illustrate modules, let’s implement a generic monad. We first define the monad interface.

```
TypeFunc "Monad"  $\Rightarrow$  (*  $\Rightarrow$  *)  $\Rightarrow$  Module
Monad 'M  $\Rightarrow$  Module {
  Func 'a  $\rightarrow$  ">>="  $\rightarrow$  ('a  $\rightarrow$  'M 'b)  $\rightarrow$  'M 'b
  Func "return"  $\rightarrow$  'a  $\rightarrow$  'M 'a
}
```

Now that we have the interface, let’s instantiate an identity monad. To use the interface, we first declare a type that fits the first argument. It has kind $(* \Rightarrow *)$ and represents the monad constructor. In case of the identity monad this is trivial.

```
TypeFunc "IdCons"  $\Rightarrow$  *  $\Rightarrow$  *
IdCons 'a  $\Rightarrow$  'a
```

The first line declares that we have a new type operator `Id` that takes a type and returns a type. The second line implements it.

To instantiate the interface we create a type operator that takes no arguments and gives us a `Monad`.

```
TypeFunc "id" => Monad
id => Monad IdCons {
  x >=> k -> k x
  return x -> x
}
```

We have now successfully implemented a monad that does nothing. While it might look useless, it is actually used quite often. When you pass an identity monad to a monad transformer, you get the corresponding monad. When you pass the identity monad to a state monad transformer, you get a state monad.

In fact, when using monad transformers, the identity monad forms the foundation of every monad. At the end of the chain of monad transformers, there is always an identity monad where all those transformations are applied to.

Inheritance

Modules can inherit from each other. This is done with the `Inherit` keyword. If -for instance- we want to extend our monad module with `Zero`, we can define a new module `MonadZero` that inherits from `Monad`.

```
TypeFunc "Zero" => Module
Zero => Module {
  Func "zero" -> 'a
}

TypeFunc "MonadZero" => (*=>*) => Module
MonadZero 'M => Module {
  Inherit Monad 'M
  Inherit Zero
}
```

This way, we have easily extended `Monad` to have a `zero`. Now `MonadZero` can behave like a monad, a 'something' with a `zero`, and a monad with a `zero`. In case of monads, having a `zero` often useful. In F# for example, defining a `zero`

for a monad means you can omit the else-branch in an if-statement.

4 Details

4.1 Meta-Casanova

```
Data "TRUE" -> Value
Data "FALSE" -> Value

Func "eval" -> Expr -> Value

Data Expr -> "&" -> Expr -> Expr
Data Expr -> "|" -> Expr -> Expr

eval a -> TRUE
eval b -> res
-----
eval (a&b) -> res

eval a -> FALSE
-----
eval (a&b) -> FALSE

eval a -> TRUE
-----
eval (a|b) -> TRUE

eval a -> FALSE
eval b -> res
-----
eval (a|b) -> res
```

Listing 1: A Boolean expression interpreter in Meta-Casanova.

Meta-Casanova uses a syntax similar to that of natural deduction. A rule is comprised of a line with below it on the left of the arrow the input, and on the right the output. Above the line are premises that are executed sequentially.

Meta-Casanova uses pattern-matching to find out what function implementation will be called.

Multiple rules may match at any given time. If this happens, the program execution splits into multiple branches. Each following one rule. If none of the rules match, the branch dies off.

This branch-and-die paradigm is very

useful for writing parsers[5]. For example: when a numeric digit is encountered, it is unclear whether an integer is being parsed or when a floating point literal is being parsed. In Meta-Casanova, you just write two rules. One that matches digits followed by a delimiter, and another that matches digits followed by a dot and optionally more digits.

Aliases

Data declarations declare a two-way many-to-1 relation between types. This two-way relationship makes Data an *alias*.

```
Data Expr → "&" → Expr → Expr
Data Expr → "|" → Expr → Expr
```

Using this excerpt from listing 1 as example, both ‘&’ and ‘|’ are defined to take two Exprs and return an Expr. They specify that a Expr can be constructed in two ways, with an ‘&’ and a ‘|’. Conversely, they also specify that an Expr can be matched in a rule with ‘&’ or ‘|’.

Premises

The statements above the horizontal line are called *premises*. They can be assignments like seen in listing 1, or conditionals like $a=3$ or $c<d$.

Often, a premise can fail. When that happens, the the entire rule fails to pattern-match and branch of execution dies. With the branch-and-die paradigm, other branches continue.

4.2 TypeFunc Details

TypeFuncs were designed to be consistent with the already existing Funcs. This has a few implications:

TypeFuncs curry

like Funcs, TypeFuncs curry.

Currying enables partially application. For example, imagine we have a state monad.

```
TypeFunc "StateMonad" ⇒ * ⇒ * ⇒ *
```

Where the first argument represents the state type and the second the return type. This means you can define a new state monad that already has its state-type specified by writing `Monad(StateMonad int)`.

It also gives the compiler a uniform way to handle type and function application, so this feature increases orthogonality while not increasing compiler complexity.

Dependent Types

Dependent types are types that depend on terms, and are currently not supported by Meta-Casanova. An example for this is the constant-length array, where the length of the array is embedded into the type at compile-time.

A future inclusion of dependent types is possible. It would mean that the type namespace will be merged into the module namespace. This won’t break backward compatibility, since all identifiers in Meta-Casanova need to be unique.

Double arrows (⇒)

We choose double arrows for TypeFuncs to clearly differentiate them from function application(\rightarrow). This distinction is to separate run-time from compile-time.

This is unlike Haskell, it uses single arrows (\rightarrow). Haskell can do this because it does not have the explicit run-time/compile-time distinction Meta-Casanova has.

Typefuncs can have premises

This means that TypeFuncs can behave just like rules. This also gives us the ability to compute arbitrary expressions at

compile-time. The only limit is that values can't be passed at compile-time, as that would require dependent types.

This is to prevent the namespace getting cluttered with dependencies.

TypeAliases

TypeAlias declarations are the mechanism for declaring type-level two-way relationships between kinds. This is analogous to the relationship Data declarations describe between types. TypeAliases enable kind-level pattern-matching in TypeFuncs, essential for making them behave like Funcs.

Modules can have premises

Modules can have premises just like Funcs and TypeFuncs. This means you can arbitrarily manipulate the input arguments for maximum expressiveness. Again the restriction here is the inability to pass term-level arguments to modules, as that would require dependent types.

4.3 Module Details

Modules were designed to be as flexible as possible, to allow them to model as great a range of abstractions as possible. This means minimizing restrictions, and maximizing orthogonality. This philosophy manifests itself in the following ways:

Scope contents

Scopes mainly consist of Func declarations, but can be used for everything. They are equivalent to top-level scope¹. We choose not to artificially limit the scope to maximize expressivity.

Modules declare new kinds

Simplest solution was to reuse the TypeFunc system for compile-time tasks. Kinds are a compile-time mechanism, and TypeFuncs are used for all compile-time tasks.

Inheritance propagation

Inherit statements inherit recursively. If C inherits from B, and B inherits from A, then C inherits both B and A. This is different from the way imports work. Import statements don't import recursively.

Inheritance vs composition

Meta-Casanova uses an inheritance-based approach as opposed to a composition-based approach.

In the inheritance-based approach, kinds inherit from each other and the users of the kinds only accept one specific kind. In the case of MonadZero, MonadZero inherits from Monad and Zero, and Funcs written for MonadZero *only* accept MonadZero or derivatives.

In the composition-based approach, kinds do not inherit from each other, and the users of the kinds accept a range of kinds. In the case of MonadZero, there would only be a Monad module and a Zero module and the Funcs that needed both would specify that in their signature.

Both strategies have their merits and are used in programming languages successfully. The advantage of a composition-based approach is that you only have to specify the building blocks, and don't have to specify all the combinations. The advantage of an inheritance-based approach is that you can add special behavior to your modules.

Because flexibility is the main concern, Meta-Casanova uses inheritance.

¹Except for import statements, those are not allowed in modules.

5 Conclusions

In this paper we described the extensions to Meta-Casanova that add modules and type operators. The extensions pave the way for modular libraries with strong abstractions and leaves room for dependent types in the future.

Related Work

To learn more about the syntax and semantics of Meta-Casanova, refer to the paper by Jarno Holstein[1]. To learn about the evolution of the standard library of Meta-Casanova and its applications, refer to the paper by Louis van der Burg[2].

For more information on Meta-Casanova, see the Meta-Casanova paper at <https://github.com/vs-team/Papers> or browse the freely-available source code at <https://github.com/vs-team/metacasanova>.

Further work

In the immediate future there will be three bachelor theses on Meta-Casanova. Jarno Holstein will write his bachelors thesis on the implementation of the type-checker for the typesystem with the extensions described in this paper. Louis van der Burg will write his on the implementation and design of the Meta-Casanova library in the Meta-Casanova language. Finally, I will write mine on the implementation of a reliable code-generator.

In the not-so-immediate future, there is the elephant in the room: full dependent types. Although dependent types complement the TypeFunc and Module system, it does not have the priority at the moment. The possibility for dependent types is therefore left open for future extensions.

References

- [1] Jarno Holstein. *Syntactic Parsing in F# Using Monadic Parsers*. 2016.
- [2] Louis van der Burg. *Adventures in Meta-Casanova*. 2016.
- [3] Bryan O’Sullivan, Don Stewart, and John Goerzen. *Real World Haskell*. <http://book.realworldhaskell.org>. O’Reilly, 2008.
- [4] Benjamin C. Pierce. *Types and Programming Languages*. The MIT Press, 2002.
- [5] Graham Hutton and Erik Meijer. *Monadic Parser Combinators*. <http://www.cs.nott.ac.uk/~pszgmh/monparsing.pdf>. University of Nottingham, 1996.